Using Procedures writing an Assembly program - Example

Dept.of CSIE

Fu Jen Catholic University, Hsin Chuang, 24205,Taipei, Taiwan. 周賜福

; **Testing the ArraySum procedure (TestArraySum.asm)**

```
.386
.model flat,stdcall
.stack 4096
ExitProcess PROTO, dwExitCode:dword

.data
array dword 10000h,20000h,30000h,40000h,50000h
theSum dword ?

.code
main proc
	mov   esi,OFFSET array		; ESI points to array
	mov   ecx,LENGTHOF array	; ECX = array count
	call  ArraySum			; calculate the sum
	mov   theSum,eax		; returned in EAX

	invoke ExitProcess,0
main endp
```

```
;-------------------------------------------------
ArraySum proc
;
; Calculates the sum of an array of 32-bit integers.
; Receives: ESI = the array offset
; ECX = number of elements in the array
; Returns: EAX = sum of the array elements
;-------------------------------------------------
      push  esi                    ; save ESI, ECX
      push  ecx
      mov   eax,0                  ; set the sum to zero

L1:
      add   eax,[esi]              ; add each integer to sum
      add   esi,TYPE DWORD         ; point to next integer
      loop  L1                     ; repeat for array size
      pop   ecx                    ; restore ECX, ESI
      pop   esi
      ret                          ; sum is in EAX

ArraySum endp
end main
```

```
; Link Library Test #2  (TestLib2.asm)

; Testing the Irvine32 Library procedures.

INCLUDE Irvine32.inc

TAB = 9         ; ASCII code for Tab

.code
main PROC
      call Randomize          ; init random generator
      call Rand1
      call Rand2
      exit
main ENDP
```

```
Rand1 PROC
; Generate ten pseudo-random integers.
      mov ecx,10          ; loop 10 times

L1:  call Random32        ; generate random int
      call WriteDec       ; write in unsigned decimal
      mov al,TAB          ; horizontal tab
      call WriteChar      ; write the tab
      loop L1

      call Crlf
      ret
Rand1 ENDP
```

```
Rand2 PROC
; Generate ten pseudo-random integers between -50 and +49
      mov ecx,10          ; loop 10 times

L1:  mov eax,100          ; values 0-99
      call RandomRange    ; generate random int
      sub  eax,50         ; values -50 to +49
      call WriteInt       ; write signed decimal
      mov al,TAB          ; horizontal tab
      call WriteChar      ; write the tab
      loop L1

      call Crlf
      ret
Rand2 ENDP

END main
```

Try to practice writing more programs in Assembly Language